Sommersemester 2015

Technische Universität Ilmenau
Institut für Mathematik
PD Dr. Erhard Hexel
Mathematik mit Maple für IN

Übungsserie 4

Lineare und affine Abbildungen, Eigenwerte und Hauptachsentransformation mit Maple

1. Beschreibung linearer Abbildungen

```
> restart
> with(LinearAlgebra) :
```

Eine **lineare Abbildung** L:$\mathbb{R}^3$->$\mathbb{R}^3$ ist etwa gegeben durch:

```
> L := (x, y, z) → Vector([2·x − z, z − x − y, x − y])
> L(x, y, z)
```

Die Bilder der Vektoren der Standardbasis {(1,0,0),(0,1,0),(0,0,1)} unter L:

```
> L(1, 0, 0); L(0, 1, 0); L(0, 0, 1);
```

Matrixdarstellung der linearen Abbildung L, das ist eine Matrix M mit L(v)=M·v:

```
> M := Matrix([L(1, 0, 0), L(0, 1, 0) , L(0, 0, 1)])
> v := Vector([x, y, z])
> 'L(v)'= Multiply(M, v)
```

Bestimmung der **darstellenden Matrix** von L mit Hilfe der Bilder einer beliebigen anderen Basis B={b1,b2,b3}, hier zum Beispiel mit b1=(1,0,0), b2=(1,1,0) und b3=(1,1,1):

```
> b1 := (1, 0, 0); b2 := (1, 1, 0); b3 := (1, 1, 1)
> A := Matrix([L(b1), L(b2) , L(b3)])
> B := Transpose(Matrix([[b1], [b2], [b3]]))
> N := Multiply(A, MatrixInverse(B))                # ein Vergleich ergibt M=N
```

Bestimmung des Kerns von L, das ist die Menge aller Vektoren v ϵ $\mathbb{R}^n$ mit L(v)=M·v=0:

```
> LinearSolve(M, ZeroVector(3), free ='t')
```

Eine **Basis des Kerns** von L:

```
> NullSpace(M)
```

<u>Man beachte:</u> Da der Kern von L hier nicht nur den Nullvektor enthält, ist L keine bijektive Abbildung. Folglich hat L keine Umkehrabbildung, das heißt die Matrix M ist nicht invertierbar:

```
> MatrixInverse(M)
```

Bestimmung der **Matrixdarstellung der Komposition** von L1 und L, wenn L1: $\mathbb{R}^3 \rightarrow \mathbb{R}^2$ die folgende lineare Abbildung ist:

```
> L1 := (x, y, z) → Vector([x + z, z − y])
> M1 := Matrix([L1(1, 0, 0), L1(0, 1, 0) , L1(0, 0, 1)])      # Matrixdarstellung von L1
> MK := Multiply(M1, M)       #Matrixdarstellung der Komposition L1 ∘L
```

Man vergleiche:

```
> L1(L(1,-1, 1)[1], L(1,-1, 1)[2], L(1,-1, 1)[3])
> Multiply(MK, Vector([1,-1, 1]))
```

2. Eigenwerte, Eigenvektoren und Eigenräume

> *restart*

> *with*(*LinearAlgebra*) :

Eine reelle (komplexe) Zahl λ heißt **Eigenwert** der reellen (komplexen) quadratischen Matrix A, wenn ein vom Nullvektor verschiedener Vektor v existiert mit Av=λv (**Eigenwertgleichung**). v heißt dann der zu λ gehörige **Eigenvektor** von A.

> $A := Matrix([[1, 2, 0], [0, 0, 0], [0, 0, 0]])$

Berechnung des charakteristischen Polynoms von A:

> $E := IdentityMatrix(3)$

> $'A - \lambda E' = A - \lambda \cdot E; \quad 'det(A - \lambda E)' = Determinant(A - \lambda \cdot E)$

> $'\lambda E - A' = \lambda \cdot E - A; \quad 'det(\lambda E - A)' = Determinant(\lambda \cdot E - A)$

Die Polynome det(A-λE) und det(λE-A) sind betragsmäßig gleich und haben deshalb die gleichen Nullstellen (das sind gerade die Eigenwerte von A). Die Berechnung des charakteristischen Polynoms von A kann einfacher mit dem *CharacteristicPolynomial*-Befehl erfolgen:

> $P := CharacteristicPolynomial(A, \text{lambda})$

Berechnung der **Eigenwerte von A** auf zwei Wegen:

> $EW := solve(P, \text{lambda})$

> $Eigenvalues(A)$ *# die Komponenten des Ergebnisvektors v sind die Eigenwerte von A*

Berechnung der **Eigenvektoren von A**:

> $L := Eigenvectors(A)$

> $'L[1]' = L[1]$ *# Vektor mit Eigenwerten von A*

> $'L[2]' = L[2]$ *# Matrix mit Eigenvektoren von A*

Der i-te Spaltenvektor von L[2] ist der zur i-ten Komponente von L[1] gehörige Eigenvektor von A:

> $L[1][1]$ *# erster Eigenwert*

> $Column(L[2], 1)$ *# zum ersten EW gehöriger Eigenvektor*

Die **Eigenräume** von A :

> $B := \lambda \rightarrow A - \lambda \cdot E$

> $'B(\lambda)' = B(\lambda)$

> $V := ZeroVector(3)$

> $LinearSolve(B(L[1][1]), V, free = 't')$ *# Eigenraum von A zu λ=L[1][1]*

> $LinearSolve(B(L[1][2]), V, free = 't')$ *# Eigenraum von A zu λ=L[1][2]*

Basen der Eigenräume:

> $NullSpace(B(L[1][1]))$

> $NullSpace(B(L[1][2]))$

3. Quadratische Formen und Hauptachsentransformation

Für die quadratische Gleichung

$$13x^2 - 10\,xy + 13y^2 = 288$$

soll eine Hauptachsentransformation durchgeführt werden.

> *restart*

> *with*(*LinearAlgebra*) : *with*(*plots*) :

> $GL := 13 \cdot x^2 - 10 \cdot x \cdot y + 13 \cdot y^2 = 288$

Graphische Darstellung der durch die Gleichung GL im $\mathbb{R}^2$ bestimmten Kurve:

> $implicitplot(GL, x = -10..10, y = -10..10)$

Matrixdarstellung der quadratischen Form $13x^2 - 10\,xy + 13y^2$:

> $A := Matrix(2, 2, [13, -5, -5, 13])$

> $z := Vector([x, y])$

> $'z^T \cdot A \cdot z' = simplify(Transpose(z) \cdot A \cdot z)$ #   *alternativ:* $simplify(BilinearForm(z, z, A, conjugate = false))$

Bestimmung der Eigenwerte und Eigenvektoren der symmetrischen Matrix A:

> $L := Eigenvectors(A)$

> $v1 := Column(L[2], 1); v2 := Column(L[2], 2)$

Bestimmung der aus den normierten Eigenvektoren von A bestehenden **Transformationsmatrix T**:

> $T := Matrix([Normalize(v1, 2), Normalize(v2, 2)])$

Die aus den Eigenwerten von A bestehende Diagonalmatrix DA (**Eigenwertmatrix**):

> $DA := DiagonalMatrix(L[1])$

Es gilt natürlich

> $'DA' = Transpose(T) \cdot A \cdot T$

> $w := Vector([u, v])$

Die ins u-v-Koordinatensystem **transformierte quadratische Gleichung** :

> $GLT := Transpose(w) \cdot DA \cdot w = 288$

Graphische Darstellung der Kurve im u-v-Koordinatensystem:

> $implicitplot(GLT, u = -10..10, v = -10..10, scaling = CONSTRAINED)$

4. Affine Abbildungen

Es bezeichne F: $\mathbb{R}^2 \to \mathbb{R}^2$ mit y=F(x) eine Spiegelung an der Geraden Γ=r0+[a] im $\mathbb{R}^2$ mit r0= $(1, 3)^T$ und a= $(2, -3)^T$ im affinen Standardkoordinatensystem (0,e1,e2).
Dann existiert eine Matrix M ϵ $\mathbb{R}^{(2,2)}$ und ein Vektor b ϵ $\mathbb{R}^2$ derart, dass für alle x ϵ $\mathbb{R}^2$

$$F(x) = M \cdot x + b$$

gilt. Matrix M und Vektor b lassen sich wie folgt berechnen:

> $restart$

> $with(LinearAlgebra):$

> $r0 := Vector([1, 3]); a := Vector([2, -3])$

Im affinen Koordinatensystem (r0,e1,e2) ist F eine Spiegelung an der Ursprungsgeraden Γ1=[a].
Das heißt, ist c ϵ $\mathbb{R}^2$ ein vom Nullvektor verschiedener Vektor mit c ⊥Γ1, dann gilt

$$M = (a, -c) \cdot (a, c)^{-1}.$$

> $c := Vector([3, 2])$

> $DotProduct(a, c)$ # *es gilt also* $c \perp a$

> $A := Matrix([a, -c]); B := Matrix([a, c])$

> $M := Multiply(A, MatrixInverse(B))$

Wegen F(x)-r0 = M·(x-r0) gilt F(x) = Mx+(r0-M·r0), also ist

$$b = r0 - M \cdot r0:$$

> $b := r0 - Multiply(M, r0)$

Aufgaben

Aufgabe 1

Gegeben seien die linearen Abbildungen F: ℝ³ ->ℝ³ mit F(x,y,z)=(x+y,2x-y,x+y+z) und G: ℝ³ -> ℝ³ mit G(x,y,z)=(z+y-x,z-2y,x-3y).
a) Bestimmen Sie die darstellenden Matrizen von F und G sowie von der Komposition G∘F.
b) Stellen Sie fest, für welche der Abbildungen F,G, G∘F eine Umkehrabbildung existiert und geben Sie diese gegebenenfalls in der obigen Schreibweise an.

Aufgabe 2

Im ℝ³ sei eine Ursprungsebene Γ=[a,b] durch die Vektoren a=$(1, 2, -1)^T$ und b=$(0, -2, 3)^T$ gegeben. Die Abbildung L: ℝ³ -> ℝ³ bezeichne die Spiegelung an Γ. Berechnen Sie mit Maple:
a) die Bilder von a,b und des Vektorproduktes von a und b unter L,
b) die Matrixdarstellung von L,
c) L(v) für v=$(-5, -7, 3)^T$ sowie den Abstand d(v,Γ).

Aufgabe 3

Gegeben sei die symmetrische Matrix

$$A = \begin{bmatrix} 1 & -2 & 5 \\ -2 & 3 & 0 \\ 5 & 0 & 2 \end{bmatrix}.$$

Bestimmen Sie eine orthogonale Matrix T derart, dass $T^T A T = T^{-1} A T$ eine Diagonalmatrix ist.

Aufgabe 4

Gegeben seien die folgenden quadratischen Gleichungen.

a) $4x^2 - 12\,xy + 3y^2 = 1$ im ℝ²,
b) $x^2 + y^2 + 5z^2 + 4\,xz = 20$ im ℝ³.

Durch die jeweilige quadratische Gleichung ist eine Kurve im ℝ² bzw. eine Fläche im ℝ³ gegeben.
Stellen Sie die Kurve/Fläche unter Anwendung des Maple-Befehls **implicitplot/ implicitplot3d** zunächst graphisch dar. Führen Sie anschließend eine Hauptachsentransformation durch und stellen Sie die Kurve/Fläche im neuen Koordinatensystem dar.

Aufgabe 5

Gegeben sei die Gerade Γ=r0+[a] im mit r0=$(-7, 5)^T$ und a=$(17, -23)^T$.
Berechnen Sie den Spiegelpunkt von x=$(11, 8)^T$ an der Geraden Γ und den Abstand von x zu Γ.